

## PRACTICAL WORK N° 2

### Numerical integration of the Schrödinger equation: vibrational energies and wave functions of the H<sub>2</sub> molecule

#### Goal :

Calculate the eigenvalues and eigenvectors (associated to the bound states) of the vibration-rotation Hamiltonian of the H<sub>2</sub> molecule. Study their behaviour in function of the input parameters.

#### I. Time-independent Schrödinger equation

Let  $\Psi(\vec{R})$  the wave function describing the relative motion of the two nuclei in a diatomic molecule, with  $\vec{R}$  denoting the vector connecting the nuclei and  $|\vec{R}| \equiv R$  being its norm. This wave function satisfies the 3-dimensional Schrödinger equation in the intramolecular potential taking account of the interactions between the nuclei and electrons.

In order to resolve this Schrödinger equation and find  $\Psi(\vec{R})$ , the separation of radial ( $R$ ) and angular ( $\theta, \varphi$ ) variables is used:

$$\Psi(\vec{R}) = \frac{1}{R} \psi(R) Y_{jm}(\theta, \varphi),$$

where  $Y_{jm}(\theta, \varphi)$  are the spherical harmonics (proper wave functions of the rotational motion) with  $J = 0, 1, \dots$  denoting the rotational quantum number (associated to the kinetic rotation moment  $\vec{J}$ ) and  $m = -J, \dots, 0, \dots, J$  standing for the magnetic quantum number (projection of  $\vec{J}$  on the z-axis of the molecule). The radial wave function  $\psi(R)$  obeys therefore the 1-dimensional Schrödinger equation:

$$H(R)\psi(R) = \left[ -\frac{\hbar^2}{2\mu} \frac{d^2}{dR^2} + V(R) + \frac{\hbar^2 J(J+1)}{2\mu R^2} \right] \psi(R) = E \psi(R),$$

where  $\mu$  is the reduced mass of the two nuclei,  $V(R)$  is the potential of the relative motion of the nuclei (vibration) and the term  $\hbar^2 J(J+1)/(2\mu R^2)$  represents the centrifugal repulsion produced by the rotation of the molecule. This equation is a homogeneous second-order differential equation, and has in consequence two linearly independent solutions. For  $R \rightarrow 0$   $V(R)$  diverges (repulsive wall due to Pauli exclusion principle), so that at the origin one of two solutions diverges whereas the second one is zero (regular solution). Because of the probabilistic interpretation of the wave function, solely the regular solution at the origin is physically acceptable. On the other side, for  $R \rightarrow \infty$   $V(R) \rightarrow 0$  and the Schrödinger equation becomes

$$-\frac{\hbar^2}{2\mu} \frac{d^2 \psi}{dR^2} = E \psi,$$

so that for  $E > 0$  one obtains oscillating (i.e. non localized in  $R$ ) solutions and for  $E < 0$  the solution is  $A \exp(-kR) + B \exp(kR)$ , with the wave number  $k = (-2\mu E / \hbar^2)^{1/2}$  and the constants  $A$  and  $B$  dependent on the initial conditions, on the shape of  $V(R)$  and on the value of  $E$ . This solution is evidently physically meaningful solely for  $B = 0$ . But the initial conditions are given ( $\psi(0) = 0$ ,  $\psi'(0) = 0$ ), the potential  $V(R)$  is given too, so that we can only play with the value of  $E$  to get  $B = 0$ . This is the quantification of the energy  $E$  values:

*the states of the molecule associated with negative energies exist only for some specific  $E$  values called "proper values of the Schrödinger equation" and these values correspond to the bound states of the molecule (in the meaning that the wave function has non negligible values solely in a limited interval of  $R$ ).*

## II. Numerical solution of the Schrödinger equation

### II.1. Reduced variables

For numerical resolving of physical problems it is very convenient to use reduced variables (without dimension). For the potential  $V(R)$  characterized by the position of the minimum  $R_0$  (distance between 2 atoms) and the potential depth  $V_0$  (roughly, the dissociation energy of the molecule), it is natural to choose  $R_0$  and  $V_0$  to “reduce” the interatomic distance and the potential itself:  $r = R/R_0$  and  $v(r) = V(R)/V_0$ . The Schrödinger equation written with these reduced variables becomes

$$\left[ -b \frac{d^2}{dr^2} + v(r) + b \frac{J(J+1)}{r^2} \right] \psi(r) = e \psi(r),$$

where  $b = B/V_0$  is the reduced value of the rotational constant  $B = \hbar^2 / (2\mu R_0^2)$  and  $e = E/V_0$  is the reduced value of the proper energy. It should be noted that for the bound states  $-1 < e < 0$ .

### II.2. Potential model

The development of numerical calculations is strongly simplified if one uses a potential model for which the analytical solutions of the Schrödinger equation are known. This is the case of the Morse potential:

$$V(R) = V_0 \left[ e^{-2a \frac{R-R_0}{R_0}} - 2e^{-a \frac{R-R_0}{R_0}} \right],$$

which gives a reasonable approximation for many diatomic molecules, at least for describing weakly excited vibrational states. In fact, this function leads to a final value for  $V(0)$ , but for realistic values of the stiffness constant  $a \gg 1$ ,  $V(0)$  is sufficiently big with respect to  $V_0$  to prevent the recovery of the electronic clouds (at least, for energies inferior or near  $V_0$ ). The use of the reduced variables this model potential is re-written as

$$v(r) = e^{-2a(r-1)} - 2e^{-a(r-1)}.$$

One can show (see lectures) that for the case  $J = 0$  the number of bound states is limited and that these levels can be labelled by the vibrational quantum number  $v$ :

$$e_v = -\left[1 - a\sqrt{b}\left(v + \frac{1}{2}\right)\right]^2, \quad 0 \leq v \leq \frac{1}{a\sqrt{b}} - \frac{1}{2}.$$

The use of this potential requires the knowledge of the parameters  $a$  et  $b$  (as well as the knowledge of  $V_0$  to come back to the energies with units). To get them, one can use experimental data on the positions of spectral lines. Namely, using the rigid rotor model (i.e. assuming that the rotational constant  $B$  entering the centrifugal term does not change from one vibrational level to another) one can write the vibrotational energy of the level  $(v, J)$  as

$$E_{v,J}^0 = E_v + BJ(J+1).$$

The spectral line positions in the branch  $S$  of the anisotropic Raman scattering spectrum of the  $H_2$  molecule for the fundamental vibration transition ( $v = 0 \rightarrow v = 1$ ) and for the first overtone ( $v = 0 \rightarrow v = 2$ ) are therefore written as

$$\nu_{0,J \rightarrow v,J+2} = E_{v,J+2}^0 - E_{0,J}^0 = E_v - E_0 + 2B(2J+3),$$

with the energies and the rotational constant expressed in wavenumber units ( $\text{cm}^{-1}$ ). Adjusting the experimentally observed line positions on this formula gives  $E_1 - E_0 = 4159,48 \text{ cm}^{-1}$ ,  $E_2 - E_0 = 8083,20 \text{ cm}^{-1}$  and  $B = 60,80 \text{ cm}^{-1}$ .

Using of these experimental values, determine the values of the parameters  $a$ ,  $b$  and  $V_0$  (before the Practical Work).


### II.3. Shooting method

The idea of this method is to choose an initial guess value for  $e$  and to correct it by an iterative procedure till obtaining an eigenvalue. It is known for example that for the Morse potential  $e$  is comprised between -1 and 0, so that to get the first vibrational level  $v = 0$  a value in the vicinity of -1 (-0,99, -0,97, etc.) should be tried. After that, to get the over levels, it will be necessary to increase this initial value of  $e$ .

Given that the solutions obtained by the direct numerical integration of the Schrödinger equation can not have a correct behaviour simultaneously for  $r \rightarrow 0$  and for  $r \rightarrow \infty$ , the shooting method is realized in two steps:

- 1) We choose a sufficiently small value of  $r = r_0$  (for the “beginning” of the wave function) in order that  $\psi(r_0) = 0$  can be assumed (the value of  $\psi'(r_0)$  can be chosen arbitrary since the wave function will be simply proportional to this value and can be re-normalized at any moment) and we integrate the Schrödinger equation (we propagate the wave function “to the right”, starting from  $r_0$  and up to some value  $r = r_m$  ( $r_m$  can be chosen for example as the first found maximum of the wave function). *Attention:  $r_0$  should not be too small to avoid the zone of the recovery of the electronic clouds !*
- 2) We choose a sufficiently big value  $r = r_N$  (the “end” of the wave function) in order to allow  $\psi(r_N) = 0$  and we integrate the Schrödinger equation (propagation of the wave function “to the left”) up to the  $r_m$  value. *Attention:  $r_N$  should not be too big to do not increase in a useless manner the time of computations!*

With a simple multiplicative factor, we can recalibrate the first (or the second) solution to get the same value for both solutions in the point  $r_m$ . We obtain thus a continuous wave function in the interval  $[r_0, r_N]$ , but the first derivative has a discontinuity in the point  $r_m$  (see the figure below) except for  $e$  being an eigenvalue.



Therefore, it is necessary to correct  $e$  by means of a correction term  $\delta e$  dependent on the discontinuity of  $\psi'$  and to re-start the both steps of the shooting method up to a convergence to a real eigenvalue.

To apply the shooting method it is necessary to know to integrate numerically the Schrödinger equation and it is also necessary to know how to calculate the correction  $\delta e$  for a given initial value of  $e$ .

### **Integration by finite differences (order $h^3$ )**

We divide the chosen  $r$  interval  $[r_0, r_N]$  into  $N$  regular subintervals of the length  $h = (r_N - r_0) / N$ . The points of the grid  $r_i$  ( $i=1,2,\dots, N$ ) can be therefore calculated as  $r_i = r_0 + ih$  and a short-hand notation  $\psi_i \equiv \psi(r_i)$  will be convenient.

Using the standard finite differences approximation

$$\psi_i'' = \frac{\psi_{i+1} + \psi_{i-1} - 2\psi_i}{h^2} + O(h^2),$$

we can see that the Schrödinger equation leads to the algorithm

$$\psi_{i+1} = \left\{ 2 + \frac{h^2}{b} \left[ b \frac{J(J+1)}{r_i^2} + v_i - e \right] \right\} \psi_i - \psi_{i-1},$$

which allows propagating  $\psi$  from  $r_0$  to  $r_m$  as soon as  $\psi_0 = 0$  and an arbitrary  $\psi_1$  are chosen.

The same algorithm enables also to propagate the wave function from  $r_N$  to  $r_m$ , if instead of looking for  $\psi_{i+1}$  we look for  $\psi_{i-1}$  starting from two previous points  $\psi_i$  and  $\psi_{i+1}$ :

$$\psi_{i-1} = \left\{ 2 + \frac{h^2}{b} \left[ b \frac{J(J+1)}{r_i^2} + v_i - e \right] \right\} \psi_i - \psi_{i+1}.$$

### **Calculation method for $\delta e$ (Cooley method)**

It is known that the functional of  $\psi$  defined by  $\langle \psi | H | \psi \rangle / \langle \psi | \psi \rangle$  equals  $E$  solely if it is an eigenvalue. Let  $\langle \psi | H | \psi \rangle / \langle \psi | \psi \rangle = E + \delta E$ , which leads to

$$\delta E = \langle \psi | H - E | \psi \rangle / \langle \psi | \psi \rangle,$$

or, numerically, to

$$\delta e = \frac{\sum_i \psi_i [-b \psi_i'' + (bJ(J+1)/r_i^2 + v_i - e) \psi_i]}{\sum_i \psi_i^2}.$$

If now we note that our method to propagate  $\psi$  makes the term [...] in the equation above always equal to zero except for the point  $r_m$  where  $\psi'$  has an discontinuity. We obtain in consequence

$$\delta e = \frac{\psi_m [-b \psi_m'' + (bJ(J+1)/r_m^2 + v_m - e) \psi_m]}{\sum_i \psi_i^2}.$$

## **II.4. Case of an arbitrary potential**

For an arbitrary potential we do not know the guess value to be given to converge to a desired  $v$  level. Consequently, it is necessary to seek for it using for example the bisection method. In this method we suppose to know two energies  $e_L$  et  $e_H$  such that the eigenvalue is between them  $e_L < e < e_H$ . So, we take as the guess value  $e = (e_L + e_H) / 2$ , we propagate the wave function and we find the number of nodes  $n_e$ . If  $n_e > v$ , we set  $e_H = e$  and if  $n_e < v$ , we set  $e_L = e$ . We repeat this procedure up to obtaining  $n_e = v$ . If we want to go closer to the eigenvalue, we can use the sign of  $\delta e$ : if  $\delta e < 0$  we set  $e_H = e$ , and if  $\delta e > 0$ , we set  $e_L = e$ , up to getting  $|\delta e / e|$  small enough to be able to start the Cooley method.

### III. Numerical studies

First, we will study the pure vibration states ( $J=0$ ) and we will seek for the eigenvalues and wave functions, starting by the ground state  $e_0$  and going up to the last level which can be detected by the method we use.


As starting values we'll take:  $r_0 = 0,1$  ;  $r_N = 2,6$  ;  $N = 50$  (i.e.  $h = 0,05$ ). We'll choose the values of  $\psi_1$  and  $\psi_{N-1}$  equal to  $\varepsilon = 10^{-4}$  and we will consider  $e$  as converged for the tolerance  $\tau = 10^{-4}$ .

Resuming the shooting method, the computation code must have the following steps:

- creation of the grid of  $r_i$  values for the interval  $[r_0, r_N]$
- calculation of  $v_i$  (these values are worthy to be saved in a vector in order to avoid to recalculate them for each iteration)
- choice of the initial guess value for  $e$
- propagation of  $\psi$  from  $r_0$  to the right up to the first found maximum  $r_m$  (with  $\psi_0 = 0$  and  $\psi_1 = \varepsilon$ )
- propagation of  $\psi$  from  $r_N$  to the left up to  $r_m$  (taking  $\psi_N = 0$  and  $\psi_{N-1} = \varepsilon$ )
- « renormalisation » of the first (or of the second) wave function to make them equal in the point  $r_m$
- calculation of  $\delta e$  ; if  $|\delta e/e| > \tau$ , we correct the energy  $e = e + \delta e$  and we restart both propagations...

Such code will be provided. Your task will be to introduce the modifications required for performing the **following studies**:

- 1) Determine numerically the vibrational energies and compare them to the theoretical values issued from the Morse potential.
- 2) For  $v = 0$  and  $v = 1$ , study the variation of the number of iterations in function of the “distance” of the guess value from the finally found eigenvalue.
- 3) Study the influence of  $r_0$ ,  $r_N$ ,  $N$ ,  $\varepsilon$  and  $\tau$  on the number of iterations and the exactness of the result for  $v = 0$  and for an highly excited vibrational level (at least  $v = 10$ ).
- 4) When the guess value is too far from the searched eigenvalue, we find another vibrational state. To understand the reason, plot a graph of variation of  $\delta e$  as a function of  $e$  for  $e$  in the interval  $[-0,98; -0,6]$ . This graph must have the following appearance:



Check that the intersections of  $\delta e$  with the  $e$  axis correspond to the eigenvalues when the derivative of  $\delta e$  is negative, and explain the problem connected to the choice of the guess value.

- 5) In order to determine the parameters of the Morse potential, the rigid rotor model was used. This model leads to an unrealistic result for the positions of the lines in the branch Q: all the lines should be on the same position  $E_v - E_0$  independently from  $J$ , whereas this fact is not verified experimentally (the lines are shifted differently from the position  $E_v - E_0$ ). We can resolve this problem by taking into account the coupling between the rotation and the vibration of the molecule; i.e. by considering the molecule as a non rigid rotor with  $B$  parametrically dependent on  $v$ :  $B_v$ .

Considering the centrifugal term as a perturbation, in the first order of the perturbation theory, the energy eigenvalues become

$$E_{v,J} = E_{v,J}^0 + BJ(J+1) \left\langle v \left| \frac{R_0^2}{R^2} \right| v \right\rangle \equiv E_v + B_v J(J+1) \text{ with } B_v \equiv B \left\langle v \left| \frac{R_0^2}{R^2} \right| v \right\rangle,$$

and the positions of the lines of the fundamental Q-branch are now given by  $E_1 - E_0 + (B_1 - B_0)J(J+1)$ . Calculate the values  $B_0$ ,  $B_1$  by the use of the wave functions found previously.

6) Compare them to the values obtained from the direct integration of the Schrödinger equation with the centrifugal term.

## PROGRAM SCHROD

```
IMPLICIT NONE
INTEGER (KIND=4), PARAMETER :: N=50
INTEGER (KIND=4):: NCMAX, IFLAG, M, JLEV, NCOUNT, NODE,I
REAL (KIND=8), DIMENSION(0:N):: V,F,RR
REAL (KIND=8):: R0,RN,A,B,EPS,TOL,FINIT,ENERG,H,HC,E,DE
DATA R0/0.1D0/, RN/2.6D0/, A/1.0/, B/1.0/, EPS/1.0D-04/, TOL/1.0D-04/,&
      FINIT/0.D0/, ENERG/-0.99D0/, NCMAX/30/, IFLAG/1/, M/11/, JLEV/0/
H=(RN-R0)/DFLOAT(N) ! calculation of the step:
HC=H*H/B ! expression h2/b
! calculation of the vector V(0:N) containing the potential
CALL POT(N,R0,H,A,B,JLEV,RR,V)
! putting the trial energy value:
E=ENERG
NCOUNT=0
WRITE(*,*) "No ITER  ENERG      M      DE      NODES"
3 CONTINUE
! propagation; NCOUNT giving the number of iterations:
CALL SHOOT(N,A,H,HC,FINIT,EPS,E,V,F,NODE,DE,M,IFLAG)
NCOUNT=NCOUNT+1
WRITE(*,200) NCOUNT, E, M, DE, NODE
200 FORMAT(I4,5X, E15.7,I4,E15.7,I4)
IF(NCOUNT.GT.NCMAX) GOTO 999
E=E+DE
IF(ABS(DE/E).GT.TOL) GOTO 3
! writing the wave function:
OPEN(UNIT=1,FILE="Wavefunction.txt")
DO I=0,N
WRITE(1,100) RR(I),F(I)
100 FORMAT(2E15.7)
ENDDO
CLOSE(1)
STOP
999 WRITE(*,*) "No convergence reached with",NCMAX," iterations"
END
```

```
SUBROUTINE POT(N,R0,H,A,B,JLEV,RR,V)
IMPLICIT NONE
INTEGER (KIND=4):: N,JLEV,I
REAL (KIND=8), DIMENSION(0:N):: V,RR
REAL (KIND=8):: R0,H,A,B
DO I=0,N
RR(I)=R0+I*H
V(I)=DEXP(-2.D0*A*(RR(I)-1.D0))-2.D0*DEXP(-A*(RR(I)-1.D0))
ENDDO
RETURN
END
```

```

SUBROUTINE SHOOT(N,A,H,HC,FINIT,EPS,E,V,F,NODE,DE,M,IFLAG)
IMPLICIT NONE
INTEGER (KIND=4):: N,NODE,M,IFLAG,ISTEP,IINIT,IFIN,I
REAL (KIND=8), DIMENSION(0:N):: V,F
REAL (KIND=8):: A,B,H,HC,FINIT,EPS,E,DE,F0,F1,COEFF,AN,FSM,FM
ISTEP=1 ! propagation do the right
IINIT=1
IFIN=N-1
IF(IFLAG.EQ.0) IFIN=M
F0=FINIT
F1=EPS
CALL PROPAG(N,HC,ISTEP,IINIT,IFIN,IFLAG,F0,F1,E,V,F,M)
FM=F(M)
ISTEP=-1 ! propagation to the left
IINIT=N-1
IFIN=M+1
F0=FINIT
F1=EPS
CALL PROPAG(N,HC,ISTEP,IINIT,IFIN,IFLAG,F0,F1,E,V,F,M)
COEFF=F(M)/FM
DO I=0,M-1
F(I)=F(I)*COEFF
ENDDO
NODE=0 ! calculation of nodes:
AN=0.D0 ! calculation of the norm for the wavefunction
DO I=1,N ! F0=0
AN=AN+F(I)*F(I)
IF(F(I)*F(I-1).LT.0.D0) NODE=NODE+1
ENDDO
FSM=F(M+1)+F(M-1)-2.D0*F(M) ! correction of the energy:
DE=F(M)*((V(M)-E)*F(M)-FSM/HC)/AN
RETURN
END

```

```

SUBROUTINE PROPAG(N,HC,ISTEP,IINIT,IFIN,IFLAG,F0,F1,E,V,F,M)
IMPLICIT NONE
INTEGER (KIND=4):: N,ISTEP,IINIT,IFIN,IFLAG,M,I,I1
REAL (KIND=8), DIMENSION(0:N):: V,F
REAL (KIND=8):: HC,F0,F1,E
F(IINIT-ISTEP)=F0
F(IINIT)=F1
DO I=IINIT,IFIN,ISTEP
I1=I+ISTEP
F(I1)=(2.0D0+HC*(V(I)-E))*F(I)-F(I-ISTEP)
IF(IFLAG.EQ.ISTEP) THEN
IF(F(I1).LT.F(I)) THEN
M=I1
RETURN
ENDIF
ENDIF
ENDDO
RETURN
END

```